



# Franck CELLIER

Formateur en Informatique

Technicien Qualité d'Organisme de Formation

<http://cellierfranck.alwaysdata.net>



## LE TABLEUR MICROSOFT EXCEL

### 14 Exemples de fonctions NB.SI

#### A - Présentation

Découverte de 16 exemples qui expliquent comment utiliser la fonction NB.SI, et ceci selon le nombre de critère utilisé et selon son type : texte, nombre, date, valeur logique ou pour identifier des doublons dans des plages de cellules. Comme nous le mentionnons toujours dans ce blog, lorsqu'une fonction comporte dans son nom le mot **SI**, cela signifie qu'elle sera **effectuée en répondant à un ou plusieurs critères** (Fonctions conditionnelles).

La **fonction NB.SI** permet donc de compter le nombre de cellules en répondant à un critère qui peut être une valeur textuelle, numérique, date ou logique ...

La syntaxe de la **fonction NB.SI** est : **NB.SI(plage; critères)**

#### B - Utiliser la fonction NB.SI avec un seul critère


##### Utilisation d'un critère textuel

Tableau de ventes réalisées par différents vendeurs :

	A	B
1	Vendeurs	Montant
2	Alex	2 565,35 €
3	Sophie	3 005,12 €
4	Jean	1 978,45 €
5	Sergio	4 615,23 €
6	Barbara	6 000,00 €
7	Younes	9 544,12 €
8	Racha	5 488,99 €
9	Sergio	3 799,78 €
10	Sophie	1 564,74 €
11	Bernard	7 456,33 €
12	Sam	5 132,55 €
13	Ana	4 012,77 €
14	Sergio	5 064,66 €
15		


Pour compter le nombre de ventes réalisées par un vendeur sélectionné.

- Choisissons un nom d'un vendeur comme exemple : **Sergio** et utilisons la fonction **NB.SI** pour effectuer le calcul souhaité.
- Sélectionnez une cellule vide et tapez la formule suivante : **=NB.SI(A2:A14;"Sergio")**
- Nous avons mis le nom **Sergio** entre guillemets puisque son type de données est texte.



	A	B	C	D	E
1	<b>Vendeurs</b>	<b>Montant</b>			
2	Alex	2 565,35 €			
3	Sophie	3 005,12 €			
4	Jean	1 978,45 €		<b>NB.SI</b>	
5	Sergio	4 615,23 €		3	
6	Barbara	6 000,00 €			
7	Younes	9 544,12 €			
8	Racha	5 488,99 €			
9	Sergio	3 799,78 €			
10	Sophie	1 564,74 €			
11	Bernard	7 456,33 €			
12	Sam	5 132,55 €			
13	Ana	4 012,77 €			
14	Sergio	5 064,66 €			

- la fonction **NB.SI** indique donc que **3 cellules** dans la plage de cellules **A2:A14** qui contiennent le nom **Sergio**.



	A	B	C	D	E
1	<b>Vendeurs</b>	<b>Montant</b>			
2	Alex	2 565,35 €			
3	Sophie	3 005,12 €			
4	Jean	1 978,45 €		<b>NB.SI</b>	
5	Sergio	4 615,23 €		3	
6	Barbara	6 000,00 €			
7	Younes	9 544,12 €			
8	Racha	5 488,99 €			
9	Sergio	3 799,78 €			
10	Sophie	1 564,74 €			
11	Bernard	7 456,33 €			
12	Sam	5 132,55 €			
13	Ana	4 012,77 €			
14	Sergio	5 064,66 €			

- Si vous remplacez **Sergio** par **Sophie**, vous allez obtenir **2**.

	A	B	C	D	E
1	<b>Vendeurs</b>	<b>Montant</b>			
2	Alex	2 565,35 €			
3	Sophie	3 005,12 €		NB.SI	
4	Jean	1 978,45 €		2	
5	Sergio	4 615,23 €			
6	Barbara	6 000,00 €			
7	Younes	9 544,12 €			
8	Racha	5 488,99 €			
9	Sergio	3 799,78 €			
10	Sophie	1 564,74 €			
11	Bernard	7 456,33 €			
12	Sam	5 132,55 €			
13	Ana	4 012,77 €			
14	Sergio	5 064,66 €			

**Note** : Vous pouvez, soit entrer le nom Sophie, soit sélectionner la cellule contenant ce nom. Votre formule pourra s'écrire de deux façons alors : **=NB.SI(A2:A14;"Sophie")** ou **=NB.SI(A2:A14;A10)**  
Remarquez également que la référence de cellule A10 n'est pas mise entre guillemets.

## Utilisation de l'astérisque

### Exemple 1 : Compter le nombre de cellules contenant des noms commençant par « S »

Sélectionnez une cellule vide et tapez la formule suivante : **=NB.SI(A2:A14;"S\*")**

Dans cet exemple l'astérisque remplace tout ce qui vient après la lettre S, et comme vous le remarquez, Excel affiche le nombre 6.

	A	B	C	D	E
1	<b>Vendeurs</b>	<b>Montant</b>			
2	Alex	2 565,35 €			
3	Sophie	3 005,12 €		NB.SI	
4	Jean	1 978,45 €		6	
5	Sergio	4 615,23 €			
6	Barbara	6 000,00 €			
7	Younes	9 544,12 €			
8	Racha	5 488,99 €			
9	Sergio	3 799,78 €			
10	Sophie	1 564,74 €			
11	Bernard	7 456,33 €			
12	Sam	5 132,55 €			
13	Ana	4 012,77 €			
14	Sergio	5 064,66 €			

## Exemple 2 : Compter le nombre de cellules contenant des noms qui se terminent par « a »

- La fonction **NB.SI** sera écrite de la façon suivante : **=NB.SI(A2:A14;"\*a")**
- Ici l'astérisque remplace tous les caractères qui se trouvent avant la lettre « a »
- Excel affiche le nombre **3**.

	A	B	C	D	E
1	Vendeurs	Montant			
2	Alex	2 565,35 €			
3	Sophie	3 005,12 €			
4	Jean	1 978,45 €		NB.SI	
5	Sergio	4 615,23 €		3	
6	Barbara	6 000,00 €			
7	Younes	9 544,12 €			
8	Racha	5 488,99 €			
9	Sergio	3 799,78 €			
10	Sophie	1 564,74 €			
11	Bernard	7 456,33 €			
12	Sam	5 132,55 €			
13	Ana	4 012,77 €			
14	Sergio	5 064,66 €			
15					

## Exemple 3 : Compter le nombre de cellules contenant un nom ou une expression...

Par exemple dans le tableau suivant, vous avez des cellules qui contiennent le nom **Rico** dont nous voulons compter leur nombre :

Sélectionnez une cellule vide et tapez la formule suivante : **=NB.SI(A2:A14;"\*rico\*")**

Nous avons mis donc le nom **Rico** entre deux astérisques parce qu'il y a des cellules où le nom complet commence par **Rico** et une cellule où **Rico** est écrit en dernier.

	A	B	C	D	E
1	Nom	Ville	Montant		
2	Alex Terrero	Paris	2 565,35 €		
3	Sophie Ioppez	Bruxelles	3 005,12 €		
4	Jean Beyan	Londres	1 978,45 €		
5	Rico Futura	Lisbone	4 615,23 €		
6	Barbara De castro	Madrir	6 000,00 €		
7	Younes Fadil	Paris	9 544,12 €		
8	Racha Bruno	Paris	5 488,99 €		
9	Rico Kloz	Lisbone	3 799,78 €		
10	Sophie Milone	Berlin	1 564,74 €		
11	Bernard Antoine	Bruxelles	7 456,33 €		
12	Sam Luna	Londres	5 132,55 €		
13	Ana Guttu	Madrir	4 012,77 €		
14	Hardy Rico	Londres	5 064,66 €		
15					
16					
17	NB.SI *Texte*	3			
18					

**Note** : Que vous écriviez « Rico » ou « rico » Excel renvoie le même résultat puisqu'on n'est pas obligé à respecter la casse en entrant les arguments de la fonction **NB.SI**.

## Utilisation des opérateurs de comparaison dans les arguments de la fonction NB.SI

### Exemple 1 : Égale à un nombre exact

À partir du tableau suivant vous allez calculer le **nombre de personnes âgées de 33 ans**:

- Sélectionnez une cellule et entrez la fonction **NB.SI** comme suit : **=NB.SI(B2:B14;33)**

	A	B	C	D
1	Nom	Age		
2	Alex Terrero	35		
3	Sophie Iopez	33		
4	Jean Beyan	33		
5	Rico Futura	41		NB.SI
6	Barbara De castro	29		4
7	Younes Fadil	34		
8	Racha Bruno	35		
9	Rico Kloz	27		
10	Sophie Milone	27		
11	Bernard Antoine	31		
12	Sam Luna	41		
13	Ana Guttu	33		
14	Hardy Rico	33		

- Vous avez donc **4 personnes qui ont l'âge de 33 ans**.

**Note** : Remarquez que le critère 33 est écrit sans guillemets parce que c'est un nombre et pas un texte.

## Supérieur ou inférieur

### Exemple 1 : Compter le nombre de cellules qui ont un montant supérieur à 1000 euros

- Entrez la formule suivante dans une cellule : **=NB.SI(C2:C14;">1000")**

	A	B	C	D
1	Nom	Date	Montant	
2	Alex Terrero	05/04/2017	2 565,35 €	
3	Sophie Iopez	03/04/2017	3 005,12 €	
4	Jean Beyan	06/04/2017	1 978,45 €	
5	Rico Futura	01/04/2017	4 615,23 €	
6	Barbara De castro	03/04/2017	6 000,00 €	
7	Younes Fadil	04/04/2017	9 544,12 €	
8	Racha Bruno	03/04/2017	5 488,99 €	
9	Rico Kloz	06/04/2017	3 799,78 €	
10	Sophie Milone	05/04/2017	1 564,74 €	
11	Bernard Antoine	03/04/2017	7 456,33 €	
12	Sam Luna	04/04/2017	5 132,55 €	
13	Ana Guttu	05/04/2017	4 012,77 €	
14	Hardy Rico	06/04/2017	5 064,66 €	
15				
16		NB.SI >1000		13

- Excel affiche **13** comme résultat de cette formule.

**Note** : Si vous avez bien remarqué, l'argument **>1000** est écrit entre guillemets. Notez donc que lorsque vous insérez des nombres accompagnés des opérateurs de comparaisons vous devez toujours les mettre entre guillemets.

## Exemple 2 : Compter le nombre de cellules contenant un montant inférieur à 2500 Euros

La fonction **NB.SI** sera utilisée de la façon suivante dans ce cas :

**=NB.SI(C2:C14;"<2500")**

	A	B	C	D
1	Nom	Date	Montant	
2	Alex Terrero	05/04/2017	2 565,35 €	
3	Sophie Iopez	03/04/2017	3 005,12 €	
4	Jean Beyan	06/04/2017	1 978,45 €	
5	Rico Futura	01/04/2017	4 615,23 €	
6	Barbara De castro	03/04/2017	6 000,00 €	
7	Younes Fadil	04/04/2017	9 544,12 €	
8	Racha Bruno	03/04/2017	5 488,99 €	
9	Rico Kloz	06/04/2017	3 799,78 €	
10	Sophie Milone	05/04/2017	1 564,74 €	
11	Bernard Antoine	03/04/2017	7 456,33 €	
12	Sam Luna	04/04/2017	5 132,55 €	
13	Ana Guttu	05/04/2017	4 012,77 €	
14	Hardy Rico	06/04/2017	5 064,66 €	
15				
16		NB.SI <2500	2	
17				

## Utilisation de la date comme critère

### Exemple 1 : Compter le nombre de cellules contenant une date précise

Dans cet exemple vous allez utiliser la fonction **NB.SI** pour trouver le nombre de cellules qui contiennent la date 06/04/2017 :


- Vous pouvez donc écrire la formule suivante : **=NB.SI(B2:B14;"06/04/2017")**
- **Attention** : Mettez le critère date entre guillemets.

	A	B	C	D
1	Nom	Date	Montant	
2	Alex Terrero	05/04/2017	2 565,35 €	
3	Sophie Iopez	03/04/2017	3 005,12 €	
4	Jean Beyan	06/04/2017	1 978,45 €	
5	Rico Futura	01/04/2017	4 615,23 €	
6	Barbara De castro	03/04/2017	6 000,00 €	
7	Younes Fadil	04/04/2017	9 544,12 €	
8	Racha Bruno	03/04/2017	5 488,99 €	
9	Rico Kloz	06/04/2017	3 799,78 €	
10	Sophie Milone	05/04/2017	1 564,74 €	
11	Bernard Antoine	03/04/2017	7 456,33 €	
12	Sam Luna	04/04/2017	5 132,55 €	
13	Ana Guttu	05/04/2017	4 012,77 €	
14	Hardy Rico	06/04/2017	5 064,66 €	
15				
16		NB.SI =06/04/2017	3	

Comme vous pouvez aussi taper la formule suivante en se servant de la **fonction DATE**

**=NB.SI(B2:B14;DATE(2017;4;6))**

Les arguments de la fonction DATE sont successivement *Année, Mois et Jour* séparés bien sûr par des points-virgules.




	A	B	C	D	E
1	<b>Nom</b>	<b>Date</b>	<b>Montant</b>		
2	Alex Terrero	05/04/2017	2 565,35 €		
3	Sophie Ioppez	03/04/2017	3 005,12 €		
4	Jean Beyan	06/04/2017	1 978,45 €		
5	Rico Futura	01/04/2017	4 615,23 €		
6	Barbara De castro	03/04/2017	6 000,00 €		
7	Younes Fadil	04/04/2017	9 544,12 €		
8	Racha Bruno	03/04/2017	5 488,99 €		
9	Rico Kloz	06/04/2017	3 799,78 €		
10	Sophie Milone	05/04/2017	1 564,74 €		
11	Bernard Antoine	03/04/2017	7 456,33 €		
12	Sam Luna	04/04/2017	5 132,55 €		
13	Ana Guttu	05/04/2017	4 012,77 €		
14	Hardy Rico	06/04/2017	5 064,66 €		
15					
16	NB.SI =06/04/2017		3		
17					

## Exemple 2 : compter le nombre de cellules contenant des dates inférieures ou égales à 05/04/2017

Entrez la **fonction NB.SI** suivante dans une cellule vide :

**=NB.SI(B2:B14;"<=05/04/2017")**



	A	B	C	D	E
1	<b>Nom</b>	<b>Date</b>	<b>Montant</b>		
2	Alex Terrero	05/04/2017	2 565,35 €		
3	Sophie Ioppez	03/04/2017	3 005,12 €		
4	Jean Beyan	06/04/2017	1 978,45 €		
5	Rico Futura	01/04/2017	4 615,23 €		
6	Barbara De castro	03/04/2017	6 000,00 €		
7	Younes Fadil	04/04/2017	9 544,12 €		
8	Racha Bruno	03/04/2017	5 488,99 €		
9	Rico Kloz	06/04/2017	3 799,78 €		
10	Sophie Milone	05/04/2017	1 564,74 €		
11	Bernard Antoine	03/04/2017	7 456,33 €		
12	Sam Luna	04/04/2017	5 132,55 €		
13	Ana Guttu	05/04/2017	4 012,77 €		
14	Hardy Rico	06/04/2017	5 064,66 €		
15					
16	NB.SI <=05/04/2017		10		



## Exemple 5 : Compter le nombre de cellules contenant des dates inférieures à la date d'aujourd'hui de 3 mois :

- Vous pouvez taper tout simplement : **=NB.SI(B2:B14;"<"&AUJOURDHUI()-90)**

	A	B	C	D	E	F
1	Nom	Date	Montant			
2	Alex Terrero	13/05/2016	2 565,35 €		Aujourd'hui	06/04/2017
3	Sophie Ioppez	02/03/2017	3 005,12 €			
4	Jean Beyan	25/01/2017	1 978,45 €		Aujourd'hui-90j	06/01/2017
5	Rico Futura	10/12/2016	4 615,23 €			
6	Barbara De castro	20/01/2017	6 000,00 €			
7	Younes Fadil	15/03/2017	9 544,12 €			
8	Racha Bruno	03/02/2017	5 488,99 €		NB.SI < Aujourd'hui-90	
9	Rico Kloz	27/10/2016	3 799,78 €		5	
10	Sophie Milone	21/11/2016	1 564,74 €			
11	Bernard Antoine	13/01/2017	7 456,33 €			
12	Sam Luna	06/02/2017	5 132,55 €			
13	Ana Guttu	08/12/2016	4 012,77 €			
14	Hardy Rico	14/02/2017	5 064,66 €			
15						

**AUJOURDHUI()-90** : renvoie la date d'aujourd'hui moins 90 jours (équivalent à 3 mois).

- Ou bien intégrez la **fonction DATE** dans la **fonction NB.SI** de la façon suivante et ceci en supposant que la date d'aujourd'hui est le **06/04/2017**

**=NB.SI(B2:B14;"<"&DATE(2017;4 ;6)-90)**

- Et si vous voulez que le calcul soit effectué automatiquement et mis à jour, entrez la formule comme suit :

**=NB.SI(B2:B14;"<"&DATE(ANNEE(AUJOURDHUI());MOIS(AUJOURDHUI())-3;JOUR(AUJOURDHUI())))**

	A	B	C	D	E	F	G
1	<b>Nom</b>	<b>Date</b>	<b>Montant</b>				
2	Alex Terrero	13/05/2016	2 565,35 €		Aujourd'hui	06/04/2017	
3	Sophie Ioppez	02/03/2017	3 005,12 €				
4	Jean Beyan	25/01/2017	1 978,45 €		Aujourd'hui- 3 mois	06/01/2017	
5	Rico Futura	10/12/2016	4 615,23 €				
6	Barbara De castro	20/01/2017	6 000,00 €				
7	Younes Fadil	15/03/2017	9 544,12 €				
8	Racha Bruno	03/02/2017	5 488,99 €		NB.SI <Aujourd'hui- 3 mois		
9	Rico Kloz	27/10/2016	3 799,78 €			5	
10	Sophie Milone	21/11/2016	1 564,74 €				
11	Bernard Antoine	13/01/2017	7 456,33 €				
12	Sam Luna	06/02/2017	5 132,55 €				
13	Ana Guttu	08/12/2016	4 012,77 €				
14	Hardy Rico	14/02/2017	5 064,66 €				

Nous avons montré précédemment que la **fonction DATE** contient trois arguments : **Année, Mois et Jour**. Voici sa syntaxe : **DATE(année;mois;jour)**

Alors, pour renvoyer l'année qui correspond à la date d'aujourd'hui nous avons utilisé une autre fonction qui est **ANNEE** en y intégrant la **fonction AUJOURDHUI()** qui renvoie elle aussi la date d'aujourd'hui : **06/04/2017**

La fonction **ANNEE(AUJOURDHUI())** est pareille donc à : **ANNEE("06/04/2017")** et renvoie **2017**.

fx		
=ANNEE(AUJOURDHUI())		
C	D	E
	2017	

Pour renvoyer le mois correspondant à la date d'aujourd'hui, nous avons utilisé la **fonction MOIS**. La formule **MOIS(AUJOURDHUI())** renvoie le **mois 4** de la date actuelle **06/04/2017**.

fx		
=MOIS(AUJOURDHUI())		
C	D	E
	4	

Et puisque nous cherchons les dates inférieures à la date d'aujourd'hui de 3 mois, nous avons tapé : **MOIS(AUJOURDHUI())-3** ce qui renvoie le **mois 1 (Janvier)**.



Il nous reste enfin de déterminer le jour de la date d'aujourd'hui, alors nous avons utilisé la **fonction JOUR** : **JOUR(AUJOURDHUI())** renvoie bien sûr 6



Alors et lorsque vous ouvrez votre classeur, votre formule destinée à **compter le nombre de cellules contenant des dates inférieures à la date d'aujourd'hui de 3 mois** sera mise à jour automatiquement et prend en compte l'actualisation de la date d'aujourd'hui !

## Utiliser la fonction NB.SI avec plusieurs critères

### Exemple 1 : Compter le nombre de cellules contenant « Sophie » et « Sergio »

Si vous reprenez le premier exemple en début de cet article, vous avez vu que nous avons entré la formule **NB.SI(A2:A14;"Sergio")** pour compter le nombre de cellules contenant « **Sergio** » et la formule **NB.SI(A2:A14;"Sophie")** pour compter le nombre de cellules contenant « **Sophie** ».

Alors pour utiliser la **fonction NB.SI** pour trouver le nombre de cellules contenant cette fois les deux noms **Sergio** et **Sophie**, il suffit tout simplement **d'additionner** les résultats de ces deux formules effectuées, comme suivant :

**= NB.SI(A2:A14;"Sergio")+ NB.SI(A2:A14;"Sophie")**

	A	B	C	D	E	F
1	<b>Vendeurs</b>	<b>Montant</b>				
2	Alex	2 565,35 €				
3	Sophie	3 005,12 €		<b>NB.SI (plusieurs critères)</b>		
4	Jean	1 978,45 €		<b>5</b>		
5	Sergio	4 615,23 €				
6	Barbara	6 000,00 €				
7	Younes	9 544,12 €				
8	Racha	5 488,99 €				
9	Sergio	3 799,78 €				
10	Sophie	1 564,74 €				
11	Bernard	7 456,33 €				
12	Sam	5 132,55 €				
13	Ana	4 012,77 €				
14	Sergio	5 064,66 €				
15						

Dans cet exemple vous avez utilisé deux critères, si vous voulez en ajouter un autre, tapez l'opérateur d'addition et entrez la **fonction NB.SI** pour le nouveau critère et ainsi de suite.

## Exemple 2 : Compter le nombre de montants supérieurs à 1000 euros et inférieurs à 3000 euros

Les deux formules que nous pouvons utiliser se sont :

- **=NB.SI(B2:B14;">1000")** qui renvoie le nombre de montants supérieurs à **1000 Euros**.
- Et **=NB.SI(B2:B14;"<3000")** qui renvoie le nombre de montants inférieurs à **3000 Euros**.

Nous allons donc utiliser ces deux formules dans une opération de soustraction mais avec une petite modification pour avoir un résultat correct :

Voici notre formule alors :

**=NB.SI(B2:B14;">1000")-NB.SI(B2:B14;">3000")**

1	Nom	Date	Montant			
2	Alex Terrero	05/04/2017	2 565,35 €			
3	Sophie Ioppez	03/04/2017	3 005,12 €			
4	Jean Beyan	06/04/2017	1 978,45 €			
5	Rico Futura	01/04/2017	4 615,23 €		NB.SI >1000 et <3000	
6	Barbara De castro	03/04/2017	6 000,00 €		3	
7	Younes Fadil	04/04/2017	9 544,12 €			
8	Racha Bruno	03/04/2017	5 488,99 €			
9	Rico Kloz	06/04/2017	3 799,78 €			
10	Sophie Milone	05/04/2017	1 564,74 €			
11	Bernard Antoine	03/04/2017	7 456,33 €			
12	Sam Luna	04/04/2017	5 132,55 €			
13	Ana Guttu	05/04/2017	4 012,77 €			
14	Hardy Rico	06/04/2017	5 064,66 €			
15						

## Compter le nombre de doublons

### Exemple 1 : Compter le nombre de doublons dans la même colonne

Une autre possibilité que fournit l'utilisation de la **fonction NB.SI** est de **trouver le nombre de doublons dans une plage de cellules**.

Reprenons encore notre exemple des noms de vendeurs, et essayons de trouver les noms similaires dans la plage **A2:A14** en utilisant **NB.SI**:

- **Sélectionnez la cellule B2** et entrez la formule suivante : **=NB.SI(\$A\$2:\$A\$14;A2)>1**
- **Si la valeur contenue dans la cellule A2 est répétée plus d'une fois dans la plage de cellules A2 :A14, Excel affichera VRAI** sinon il affichera FAUX.
- Copiez la formule dans les cellules de **B2 à B14**,

et voici ce que vous allez obtenir :

	A	B	C	D	E
1	Vendeurs	NB.SI Doublons			
2	Alex	FAUX			
3	Sophie	VRAI			
4	Jean	FAUX			
5	Sergio	VRAI			
6	Barbara	FAUX			
7	Younes	FAUX			
8	Racha	FAUX			
9	Sergio	VRAI			
10	Sophie	VRAI			
11	Bernard	FAUX			
12	Sam	FAUX			
13	Ana	FAUX			
14	Sergio	VRAI			
15					

- Pour calculer ensuite le nombre de cellules contenant les doublons, entrez la formule suivante :

**=NB.SI(B2:B14;VRAI)**

	A	B	C	D
1	Vendeurs	NB.SI Doublons		
2	Alex	FAUX		
3	Sophie	VRAI		
4	Jean	FAUX	Nbre doublons	
5	Sergio	VRAI	5	
6	Barbara	FAUX		
7	Younes	FAUX		
8	Racha	FAUX		
9	Sergio	VRAI		
10	Sophie	VRAI		
11	Bernard	FAUX		
12	Sam	FAUX		
13	Ana	FAUX		
14	Sergio	VRAI		
15				

**Note :** Remarquez que la valeur logique **VRAI** est saisie sans guillemets.

Il y a une autre solution pour ce faire également, qui consiste à utiliser **NB.SI** intégrée dans la fonction **SOMMEPROD**.

Voici comment procéder :

- Sélectionnez une cellule vide et entrez la formule suivante :

**=SOMMEPROD((NB.SI(A2:A14;A2:A14)>1)\*(A2:A14<>""))**

	A	C	D	E
1	Vendeurs			
2	Alex			
3	Sophie			
4	Jean	Nbre doublons		
5	Sergio		5	
6	Barbara			
7	Younes			
8	Racha			
9	Sergio			
10	Sophie			
11	Bernard			
12	Sam			
13	Ana			
14	Sergio			
15				

- Le résultat est le même : **5 doublons** aussi.

**Note :** pour plus de détails sur l'utilisation de cette syntaxe de la fonction SOMMEPROD, visitez le lien suivant : [Moyenne pondérée avec condition](#)

## Exemple 2 : Compter le nombre de doublons dans deux colonnes

Les **deux colonnes A et B** contiennent des noms différents, cependant il y en a quelques-uns qui sont similaires dans les deux colonnes :

- Entrez la formule suivante pour calculer le nombre de ces doublons :

**=SOMMEPROD((NB.SI(A1:A13;B1:B13)>0)\*(B1:B13<>""))**

	A	B	C	D	E	F
1	Alex	Alex				
2	Sophie	Ioppez				
3	Jean	Jean				
4	Sergio	Futura	NB.SI doublons dans les deux colonnes			
5	Barbara	Barbara				
6	Younes	Younes			8	
7	Racha	Racha				
8	Sergio	Kloz				
9	Sophie	Milone				
10	Bernard	Bernard				
11	Sam	Sam				
12	Ana	Ana				
13	Sergio	Rico				
14						

- **NB.SI(A1:A13;B1:B13)>0** : l'utilisation de **NB.SI** dans ce cas permet de calculer le nombre de noms de la plage **A1:A13** qui se répètent au moins une fois dans la plage **B1:B13**.

- Et pour que le calcul s'effectue sans erreur, la plage **B1:B13** ne doit pas être vide, c'est pourquoi que nous avons entré ce test logique **B1:B13<>"**.